

Bewegung in homogenen elektromagnetischen Feldern

Hendrik van Hees
04. Juni 2000

■ Allgemeine Vorbemerkungen

Das elektromagnetische Feld ist in der kovarianten Schreibweise ein antisymmetrischer Tensor, der Faradaytensor. Im folgenden sind alle Vektoren und Tensoren (es kommen nur solche bis zur 2. Stufe vor) mit Indizes oben (kontravariant) zu verstehen. Wir arbeiten mit dem in der Teilchenphysik weitgehend üblichen rationalisierten gaußschen Maßsystem mit $c=1$ (Heaviside units). Wir folgen in der Definition der relativistischen kovarianten Schreibweise wie sie in Bjorken–Drell steht (was sich auch weitgehend durchgesetzt hat). Auch das Particle Data Booklet hält sich an diese Konvention. Der Fundamentaltensor (Pseudometrik) ist also

```
In[1]:= g = {{1, 0, 0, 0}, {0, -1, 0, 0}, {0, 0, -1, 0}, {0, 0, 0, -1}};
```

```
In[2]:= MatrixForm[g]
```

```
Out[2]//MatrixForm=  

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{pmatrix}$$

```

Der Faradaytensor für ein beliebiges elektromagnetisches Feld liest sich

```
In[3]:= Evec = {E1, E2, E3}; Bvec = {B1, B2, B3};
```

```
General::spell1 :  
Possible spelling error: new symbol name "Bvec" is similar to existing symbol "Evec".
```

```
In[4]:= F[{E1_, E2_, E3_}, {B1_, B2_, B3_}] =  
{{0, -E1, -E2, -E3}, {E1, 0, -B3, B2}, {E2, B3, 0, -B1}, {E3, -B2, B1, 0}};
```

```
In[5]:= MatrixForm[F[Evec, Bvec]]
```

```
Out[5]//MatrixForm=  

$$\begin{pmatrix} 0 & -E1 & -E2 & -E3 \\ E1 & 0 & -B3 & B2 \\ E2 & B3 & 0 & -B1 \\ E3 & -B2 & B1 & 0 \end{pmatrix}$$

```

Die Bewegungsgleichung wird in vierdimensional kovarianter Gestalt mit der Eigenzeit τ (eine spezielle Wahl eines Weltzeitparameters) geschrieben. Die kovariante Geschwindigkeit heißt $u=dx/d(\tau)$.

```
In[6]:= u[tau_] = {u0[tau], u1[tau], u2[tau], u3[tau]};
```

Die Bewegungsgleichung lautet (für eine Herleitung vgl. z.B. Landau–Lifshitz Vol. 2)

```
In[7]:= eom = Table[mD[u[tau], tau][[i]] == q (F[Evec, Bvec] . g . u[tau])[[i]], {i, 1, 4}];
```

```
In[8]:= MatrixForm[eom]
```

```
Out[8]//MatrixForm=
```

$$\begin{pmatrix} m u_0'[\tau] == q (E_1 u_1[\tau] + E_2 u_2[\tau] + E_2 u_3[\tau]) \\ m u_1'[\tau] == q (E_1 u_0[\tau] + B_3 u_2[\tau] - B_2 u_3[\tau]) \\ m u_2'[\tau] == q (E_2 u_0[\tau] - B_3 u_1[\tau] + B_1 u_3[\tau]) \\ m u_3'[\tau] == q (E_2 u_0[\tau] + B_2 u_1[\tau] - B_1 u_2[\tau]) \end{pmatrix}$$

Darin steht m für die invariante Masse. Eine geschwindigkeitsabhängige Masse, wie sie in älteren Lehrbüchern steht, ist wenig hilfreich, da sie keine relativistisch kovariante Größe ist. Die rechte Seite der Gleichung ist in den drei Raumkomponenten unschwer als die Lorentzkraft in vierdimensional kovarianter Gestalt zu erkennen. Die Bewegungsgleichung ist mit der Bedingung, daß τ die Eigenzeit ist, verträglich, denn es gilt

```
In[9]:= u[\tau] . g . (F[Evec, Bvec] . g . u[\tau]) // Simplify
```

```
Out[9]= 0
```

Das bedeutet, daß $u.g.u=const.$ ist. Die Eigenzeit wählen wir nun so, daß sie die Zeit im Ruhssystem des Betrachters ist, so daß $u.g.u=1$ sein muß. Die Bewegung besitzt also eine Zwangsbedingung 2. Klasse (d.h. eine solche, die nur bei Erfüllung der Bewegungsgleichung gelten muß, aber das ist für die klassische Physik wenig wichtig, erst beim Formalismus der 1. Quantisierung wird das wichtig).

■ Bewegung im homogenen elektrischen Feld

In diesem Abschnitt soll die Bewegung eines Punktteilchens im homogenen elektrischen Feld dargestellt werden. O.b.d.A. legen wir das elektrische Feld in Richtung der z -Achse.

```
In[10]:= MatrixForm[F[{0, 0, Ez}, {0, 0, 0}]]
```

```
Out[10]//MatrixForm=
```

$$\begin{pmatrix} 0 & 0 & 0 & -Ez \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ Ez & 0 & 0 & 0 \end{pmatrix}$$

```
In[11]:= g1 =
```

```
Table[mD[u[\tau], \tau][[i]] == q (F[{0, 0, Ez}, {0, 0, 0}] . g . u[\tau])[[i]], {i, 1, 4}];
```

```
In[12]:= MatrixForm[g1]
```

```
Out[12]//MatrixForm=
```

$$\begin{pmatrix} m u_0'[\tau] == Ez q u_3[\tau] \\ m u_1'[\tau] == 0 \\ m u_2'[\tau] == 0 \\ m u_3'[\tau] == Ez q u_0[\tau] \end{pmatrix}$$

```
In[13]:= solution = DSolve[g1, {u0[\tau], u1[\tau], u2[\tau], u3[\tau]}, \tau] // FullSimplify
```

```
Out[13]= {{u0[\tau] -> C[1] Cosh[\frac{Ez q \tau}{m}] + C[4] Sinh[\frac{Ez q \tau}{m}], u1[\tau] -> C[2],
u2[\tau] -> C[3], u3[\tau] -> C[4] Cosh[\frac{Ez q \tau}{m}] + C[1] Sinh[\frac{Ez q \tau}{m}]}}
```

In dieser Lösung ist die Zwangsbedingung 2. Klasse zu berücksichtigen

```
In[14]:= u[\tau] . g . u[\tau] /. Flatten[solution] // FullSimplify
```

```
Out[14]= C[1]^2 - C[2]^2 - C[3]^2 - C[4]^2
```

D.h. es muß $C[1]^2=1+C[2]^2+C[3]^2+C[4]^2$ gelten. Damit weiter u_0 stets positiv ist (Bewegung in positive Zeitrichtung muß hier von Hand eingesetzt werden \rightarrow Bewegungsumkehrinvarianz der elektromagnetischen

Wechselwirkung. Die Gleichung zeichnet keine bestimmte Zeitrichtung aus). Die freien Integrationskonstanten C[2], C[3], C[4] sind entsprechend der Anfangsbedingungen für die Vierergeschwindigkeit zu wählen.

```
In[15]:= usol[tau_] =
          (u[tau] /. Flatten[solution]) /. {C[1] -> Sqrt[1 + C[2]^2 + C[3]^2 + C[4]^2]};
```

```
In[16]:= MatrixForm[usol[tau]]
```

```
Out[16]//MatrixForm=
```

$$\begin{pmatrix} \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \operatorname{Cosh}\left[\frac{Ez q \tau}{m}\right] + C[4] \operatorname{Sinh}\left[\frac{Ez q \tau}{m}\right] \\ C[2] \\ C[3] \\ C[4] \operatorname{Cosh}\left[\frac{Ez q \tau}{m}\right] + \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \operatorname{Sinh}\left[\frac{Ez q \tau}{m}\right] \end{pmatrix}$$

Die Lösung für die Raumzeitkoordinaten ergibt sich durch Integration. Wir legen o.B.d.A. den Anfangspunkt der Bewegung in den Ursprung des Koordinatensystems

```
In[17]:= x[tau_] = Integrate[usol[tauPrime], {tauPrime, 0, tau}];
```

```
In[18]:= MatrixForm[x[tau]]
```

```
Out[18]//MatrixForm=
```


$$\begin{pmatrix} -\frac{m C[4]}{Ez q} + \frac{m C[4] \operatorname{Cosh}\left[\frac{Ez q \tau}{m}\right]}{Ez q} + \frac{m \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \operatorname{Sinh}\left[\frac{Ez q \tau}{m}\right]}{Ez q} \\ \tau C[2] \\ \tau C[3] \\ -\frac{m \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2}}{Ez q} + \frac{m \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \operatorname{Cosh}\left[\frac{Ez q \tau}{m}\right]}{Ez q} + \frac{m C[4] \operatorname{Sinh}\left[\frac{Ez q \tau}{m}\right]}{Ez q} \end{pmatrix}$$

Jetzt zeigen wir exemplarisch ein paar Ergebnisse

```
In[19]:= Ez = 1; q = 1; m = 1;
```

Die Geschwindigkeit v3 in z-Richtung für einen im benutzten Koordinatensystem ruhenden Beobachter ist u3/u0, und natürlich nähert sich diese Geschwindigkeit der Lichtgeschwindigkeit c=1 und überschreitet diese nie:


```
In[20]:= Plot[usol[tau][[4]] / usol[tau][[1]] /. {C[2] -> 0, C[3] -> 0, C[4] -> 0}, {tau, 0, 5},
          PlotRange -> {0, 1}, AxesLabel -> {"tau", "v3"}, PlotStyle -> AbsoluteThickness[2]]
```



```
Out[20]= - Graphics -
```

Die Bahnkurve sieht folgendermaßen aus. Hier starten wir mit einer gewissen Anfangsgeschwindigkeit in x-Richtung. Das entspricht Anfangsbedingungen des "waagrechten Wurfes" im nichtrelativistischen Fall wäre das eine Parabel


```
In[21]:= ParametricPlot[
  Evaluate[{x[tau][[2]], x[tau][[4]]} /. {C[2] -> 0.2, C[3] -> 0, C[4] -> 0}], {tau, 0, 2},
  AxesLabel -> {"x", "z"}, PlotStyle -> AbsoluteThickness[2], AxesLabel -> {"x", "z"}]
```



Out[21]= - Graphics -

Für eine Anfangsbedingung, die einem schrägen Wurf mit einer Anfangsgeschwindigkeits- z -Komponente in Gegenrichtung zum E-Feld entspricht, sieht das Bild so aus:

```
In[22]:= ParametricPlot[
  Evaluate[{x[tau][[2]], x[tau][[4]]} /. {C[2] -> 0.2, C[3] -> 0, C[4] -> -0.6}],
  {tau, 0, 2}, PlotStyle -> AbsoluteThickness[2], AxesLabel -> {"x", "z"}]
```



Out[22]= - Graphics -

■ Bewegung im konstanten Magnetfeld

```
In[23]:= Clear[q, m, Ez]
```

In diesem Abschnitt soll die Bewegung eines Punktteilchens im homogenen magnetischen Feld dargestellt werden. O.b.d.A. legen wir das elektrische Feld in Richtung der z -Achse.

In[24]:= MatrixForm[F[{0, 0, 0}, {0, 0, Bz}]]

Out[24]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & -Bz & 0 \\ 0 & Bz & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

In[25]:= g1 =
 Table[mD[u[tau], tau][[i]] == q (F[{0, 0, 0}, {0, 0, Bz}] . g . u[tau])[[i]], {i, 1, 4}];

In[26]:= MatrixForm[g1]

Out[26]//MatrixForm=

$$\begin{pmatrix} m u_0'[\tau] == 0 \\ m u_1'[\tau] == Bz q u_2[\tau] \\ m u_2'[\tau] == -Bz q u_1[\tau] \\ m u_3'[\tau] == 0 \end{pmatrix}$$

In[27]:= solution = DSolve[g1, {u0[tau], u1[tau], u2[tau], u3[tau]}, tau] // FullSimplify

Out[27]= {{u0[tau] -> C[1], u1[tau] -> C[2] Cos[$\frac{Bz q \tau}{m}$] + C[3] Sin[$\frac{Bz q \tau}{m}$],
 u2[tau] -> C[3] Cos[$\frac{Bz q \tau}{m}$] - C[2] Sin[$\frac{Bz q \tau}{m}$], u3[tau] -> C[4]}}

In dieser Lösung ist die Zwangsbedingung 2. Klasse zu berücksichtigen

In[28]:= u[tau] . g . u[tau] /. Flatten[solution] // FullSimplify

Out[28]= C[1]^2 - C[2]^2 - C[3]^2 - C[4]^2

D.h. es muß $C[1]^2 = 1 + C[2]^2 + C[3]^2 + C[4]^2$ gelten. Damit weiter u_0 stets positiv ist (Bewegung in positive Zeitrichtung muß hier von Hand eingesetzt werden -> Bewegungsumkehrinvarianz der elektromagnetischen Wechselwirkung. Die Gleichung zeichnet keine bestimmte Zeitrichtung aus). Die freien Integrationskonstanten $C[2]$, $C[3]$, $C[4]$ sind entsprechend der Anfangsbedingungen für die Vierergeschwindigkeit zu wählen.

In[29]:= usol[tau_] =
 (u[tau] /. Flatten[solution]) /. {C[1] -> Sqrt[1 + C[2]^2 + C[3]^2 + C[4]^2]};

In[30]:= MatrixForm[usol[tau]]

Out[30]//MatrixForm=

$$\begin{pmatrix} \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \\ C[2] \cos\left[\frac{Bz q \tau}{m}\right] + C[3] \sin\left[\frac{Bz q \tau}{m}\right] \\ C[3] \cos\left[\frac{Bz q \tau}{m}\right] - C[2] \sin\left[\frac{Bz q \tau}{m}\right] \\ C[4] \end{pmatrix}$$

Die Lösung für die Raumzeitkoordinaten ergibt sich durch Integration. Wir legen o.B.d.A. den Anfangspunkt der Bewegung in den Ursprung des Koordinatensystems

In[31]:= x[tau_] = Integrate[usol[tauPrime], {tauPrime, 0, tau}];

In[32]:= MatrixForm[x[tau]]

Out[32]//MatrixForm=


$$\begin{pmatrix} \tau \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \\ \frac{m C[3]}{Bz q} - \frac{m C[3] \cos\left[\frac{Bz q \tau}{m}\right]}{Bz q} + \frac{m C[2] \sin\left[\frac{Bz q \tau}{m}\right]}{Bz q} \\ - \frac{m C[2]}{Bz q} + \frac{m C[2] \cos\left[\frac{Bz q \tau}{m}\right]}{Bz q} + \frac{m C[3] \sin\left[\frac{Bz q \tau}{m}\right]}{Bz q} \\ \tau C[4] \end{pmatrix}$$

Jetzt zeigen wir exemplarisch ein paar Ergebnisse

```
In[33]:= Bz = 1; q = 1; m = 1;
```

Die Geschwindigkeitskomponente v_1 in x -Richtung für einen im benutzten Koordinatensystem ruhenden Beobachter ist u_1/u_0 . Die Lösung hat oszillatorischen Charakter. Der Grund ist klar: Die Projektion der Bahnkurve auf die xy -Ebene ist ein Kreis. Die Gesamtbewegung i.a. eine Spiralbahn (s. u.). In diesem Fall ist, wie ein Blick auf die Lösung zeigt, die Eigenzeit proportional zur Systemzeit des Beobachters.


```
In[34]:= Plot[usol[tau][[2]] / usol[tau][[1]] /. {C[2] -> 0.2, C[3] -> 0, C[4] -> 0},  
  {tau, 0, 10}, AxesLabel -> {"tau", "x"}, PlotStyle -> AbsoluteThickness[2]]
```



```
Out[34]= - Graphics -
```

Die Bahnkurve in xy -Ebene sieht folgendermaßen aus. Hier starten wir mit einer gewissen Anfangsgeschwindigkeit in x -Richtung. In der Tat beschreibt das Teilchen eine Kreisbahn, wie entsprechend der Lorentzkraft im reinen Magnetfeld zu erwarten war: Die Kraft wirkt in jedem Moment senkrecht zur Geschwindigkeit!


```
In[35]:= ParametricPlot[Evaluate[
  {x[tau][[2]], x[tau][[3]]} /. {C[2] → -0.2, C[3] → 0, C[4] → 0}], {tau, 0, 7},
  AspectRatio → 1, AxesLabel → {"x", "y"}, PlotStyle → AbsoluteThickness[2]]
```



```
Out[35]= - Graphics -
```

Für eine allgemeine Anfangsbedingung lohnt sich ein 3-dim-Plot, denn i.a. ist die Bahnkurve eine Schraubenlinie konstanter Ganghöhe (die Geschwindigkeit in Richtung des Magnetfeldes bleibt konstant, weil die Kraft stets senkrecht zur Geschwindigkeit und zum Magnetfeld wirkt $F_{\text{mag}}=q \mathbf{v} \times \mathbf{B}$).

```
In[36]:= ParametricPlot3D[Evaluate[
  {x[tau][[2]], x[tau][[3]], x[tau][[4]]} /. {C[2] -> 0.5, C[3] -> 0, C[4] -> 0.1}],
  {tau, 0, 20}, AspectRatio -> 1, ViewPoint -> {2.895, 1.108, 1.356},
  AxesLabel -> {"x", "y", "z"}]
```



Out[36]= - Graphics3D -

■ Bewegung im gekreuzten E- und B-Feld

```
In[37]:= Clear[q, m, Ez, Bz]
```

In diesem Abschnitt soll die Bewegung eines Punktteilchens im homogenen elektrischen Feld (in y-Richtung) magnetischem Feld (in z-Richtung) dargestellt werden.

```
In[38]:= MatrixForm[F[{0, Ey, 0}, {0, 0, Bz}]]
```

```
Out[38]//MatrixForm=
  (  0  0  -Ey  0
    0  0  -Bz  0
    Ey Bz  0  0
    0  0  0  0 )
```

```
In[39]:= g1 = Table[mD[u[tau], tau][[i]] == q (F[{0, Ey, 0}, {0, 0, Bz}] . g.u[tau])[i],
  {i, 1, 4}];
```

```
In[40]:= MatrixForm[g1]
```

```
Out[40]//MatrixForm=
  (  m u0'[tau] == Ey q u2[tau]
    m u1'[tau] == Bz q u2[tau]
    m u2'[tau] == q (Ey u0[tau] - Bz u1[tau])
    m u3'[tau] == 0 )
```

```
In[41]:= solution = DSolve[g1, {u0[tau], u1[tau], u2[tau], u3[tau]}, tau] // FullSimplify;
```


In dieser Lösung ist die Zwangsbedingung 2. Klasse zu berücksichtigen

```
In[42]:= u[tau] . g . u[tau] /. Flatten[solution] // FullSimplify
```

```
Out[42]= C[1]^2 - C[2]^2 - C[3]^2 - C[4]^2
```

D.h. es muß $C[1]^2=1+C[2]^2+C[3]^2+C[4]^2$ gelten. Damit weiter u_0 stets positiv ist (Bewegung in positive Zeitrichtung muß hier von Hand eingesetzt werden→Bewegungsumkehrinvarianz der elektromagnetischen Wechselwirkung. Die Gleichung zeichnet keine bestimmte Zeitrichtung aus). Die freien Integrationskonstanten $C[2]$, $C[3]$, $C[4]$ sind entsprechend der Anfangsbedingungen für die Vierergeschwindigkeit zu wählen.

```
In[43]:= usol[tau_] =
(u[tau] /. Flatten[solution]) /. {C[1] -> Sqrt[1 + C[2]^2 + C[3]^2 + C[4]^2]};
```

```
In[44]:= usol[tau]
```

$$\text{Out[44]} = \left\{ \frac{1}{(Bz - Ey)(Bz + Ey)} \left(Bz \left(-Ey C[2] + Bz \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \right) - \right. \right. \\ \left. \left. Ey \left(-Bz C[2] + Ey \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \right) \cosh \left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m} \right] - \right. \right. \\ \left. \left. Ey \sqrt{-Bz^2 + Ey^2} C[3] \sinh \left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m} \right] \right), \right. \\ \frac{1}{(Bz - Ey)(Bz + Ey)} \left(Ey \left(-Ey C[2] + Bz \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \right) + \right. \\ \left. Bz \left(Bz C[2] - Ey \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \right) \cosh \left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m} \right] - \right. \\ \left. Bz \sqrt{-Bz^2 + Ey^2} C[3] \sinh \left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m} \right] \right), \\ C[3] \cosh \left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m} \right] + \\ \frac{\left(-Bz C[2] + Ey \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \right) \sinh \left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m} \right]}{\sqrt{-Bz^2 + Ey^2}}, \\ \left. C[4] \right\}$$

Die Lösung für die Raumzeitkoordinaten ergibt sich durch Integration. Wir legen o.B.d.A. den Anfangspunkt der Bewegung in den Ursprung des Koordinatensystems

```
In[45]:= x[tau_] = Integrate[usol[tauPrime], {tauPrime, 0, tau}];
```

Leider erlaubt das Exportieren nach ps bzw. html nicht die Benutzung von MatrixForm für so lange Ausdrücke, weil die Ausgabe einfach abgeschnitten wird: ;-(

In[46] := x[tau]


$$\text{Out[46]} = \left\{ -\frac{Bz Ey \tau C[2]}{(Bz - Ey)(Bz + Ey)} + \frac{Ey m C[3]}{(Bz - Ey)(Bz + Ey) q} + \frac{Bz^2 \tau \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2}}{(Bz - Ey)(Bz + Ey)} - \frac{Ey m C[3] \text{Cosh}\left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m}\right]}{(Bz - Ey)(Bz + Ey) q} + \frac{Bz Ey m C[2] \text{Sinh}\left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m}\right]}{(Bz - Ey)(Bz + Ey) \sqrt{-Bz^2 + Ey^2} q} - \frac{Ey^2 m \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \text{Sinh}\left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m}\right]}{(Bz - Ey)(Bz + Ey) \sqrt{-Bz^2 + Ey^2} q} - \frac{Ey^2 \tau C[2]}{(Bz - Ey)(Bz + Ey)} + \frac{Bz m C[3]}{(Bz - Ey)(Bz + Ey) q} + \frac{Bz Ey \tau \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2}}{(Bz - Ey)(Bz + Ey)} - \frac{Bz m C[3] \text{Cosh}\left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m}\right]}{(Bz - Ey)(Bz + Ey) q} + \frac{Bz^2 m C[2] \text{Sinh}\left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m}\right]}{(Bz - Ey)(Bz + Ey) \sqrt{-Bz^2 + Ey^2} q} - \frac{Bz Ey m \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \text{Sinh}\left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m}\right]}{(Bz - Ey)(Bz + Ey) \sqrt{-Bz^2 + Ey^2} q}, \frac{1}{(Bz^2 - Ey^2) q} \left(-Bz m C[2] + Ey m \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} + Bz m C[2] \text{Cosh}\left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m}\right] - Ey m \sqrt{1 + C[2]^2 + C[3]^2 + C[4]^2} \text{Cosh}\left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m}\right] - \sqrt{-Bz^2 + Ey^2} m C[3] \text{Sinh}\left[\frac{\sqrt{-Bz^2 + Ey^2} q \tau}{m}\right] \right), \tau C[4] \right\}$$

Jetzt zeigen wir exemplarisch ein paar Ergebnisse

In[47] := Ey = 0.5; Bz = 1; q = 1; m = 1;

Die Geschwindigkeitskomponente v1 in x-Richtung für einen im benutzten Koordinatensystem ruhenden Beobachter ist u1/u0. Die Lösung hat oszillatorischen Charakter. Der Grund ist klar: Die Projektion der Bahnkurve auf die xy-Ebene ist ein Kreis. Die Gesamtbewegung i.a. eine Spiralbahn (s. u.). In diesem Fall ist, wie ein Blick auf die Lösung zeigt, die Eigenzeit proportional zur Systemzeit des Beobachters.


In[48] := Plot[usol[tau][[2]]/usol[tau][[1]] /. {C[2] -> 0.2, C[3] -> 0, C[4] -> 0}, {tau, 0, 10}, AxesLabel -> {"tau", "x"}, PlotStyle -> AbsoluteThickness[2]]



Out[48] = - Graphics -

Die Bahnkurve in xy-Ebene sieht folgendermaßen aus. Hier starten wir mit einer gewissen Anfangsgeschwindigkeit in x-Richtung.

```
In[49]:= ParametricPlot[Evaluate[
  {x[tau][[2]], x[tau][[3]]} /. {C[2] → -0.2, C[3] → 0, C[4] → 0}], {tau, 0, 20},
  AspectRatio → 1, AxesLabel → {"x", "y"}, PlotStyle → AbsoluteThickness[2]]
```




Out[49]= - Graphics -

Für eine allgemeine Anfangsbedingung lohnt sich ein 3-dim-Plot. Die Anfangsbedingungen wurden so gewählt, daß eine einigermaßen ansehnliche Bahnkurve entsteht. Besitzer von *Mathematica* sollten ein wenig mit diesem Notebook spielen ;-)).

```
In[50]:= Sqrt[3 * 0.3^2]
```

Out[50]= 0.519615

```
In[51]:= ParametricPlot3D[Evaluate[
  {x[tau][[2]], x[tau][[3]], x[tau][[4]]} /. {C[2] → -0.01, C[3] → 0.5, C[4] → 0.3},
  {tau, 0, 20}, AxesLabel → {"x", "y", "z"}, ViewPoint -> {2.885, -1.269, 1.231}]
```



Out[51]= - Graphics3D -